
Using TF Macro Tool
The TF Macro Tool permits “TeleFinder/User” to establish an asynchronous serial connection by 
bypassing TeleFinder’s internal connection script.  The Macro Tool allows the user to script his own 
connection routine with a  Macro Script.

The Macro Script can be written with any text editor that can save a file with a creator of ‘TFMC’.  
TeleFinder’s built-in editor will do this when creating a text file, then selecting “Save As”.  In the save 
dialog, click in the “Other” radio button and enter “TFMC” into the creator field to the right of the 
button, then save the file.

TF Macro Tool Commands
The commands you can script with are as follows:

control of serial port
BAUD
SETCOMM
FLOW
REMECHO
LOCECHO
BREAK

sending data
TYPE
SLOW

receiving and comparing data
PROMPT
PROMPSEC
GETLINE
CMPLINE

controlling macro execution
PAUSE
GOTO
TRUE
FALSE
END

setting flags and misc.
CONNECT
CANCEL
BELL
*

Macro File Format
The macro file is an ascii text file.  The Macro Tool parses the file at run time.  Commands must begin
on the first character of a line, followed by its arguments separated by a SPACE or TAB.  Character 
parameters are enclosed in quotes and numeric arguments are terminated by a semi-colon.  Each 
statement line is terminated by a CR.  Commands are always in upper case.

sample format

BAUD 2400;
SLOW 4;
TYPE "AT^M"
PROMPSEC 4;
PROMPT "OK"

The sample above sets the baud rate to 2400 bps.  It then sets the  typing delay to 4 ticks  (1/60 sec).
Then it types AT<CR>.  The macro then waits up to 4 seconds to receive “OK” .

Typical Macro Shell
The typical macro will set up a connection, then report back to TeleFinder/User the state of the 
connection, and whether or not to proceed with logon to the host.  The Flag commands (CONNECT 
and CANCEL ) communicate this to TeleFinder.

This example leaves out  the statements necessary to establish a connection, those will vary from 
device to device.  With every macro the ending will be very similar, branch to a label that reports 
connection status.  Here we use “connectExit” when a connection was successfully established and 
“errorExit” when we could not make  a connection.

shell example
* TF Macro Tool script example ( lines beginning with * are comments )
* statements used to establish a connection follow...
*
* (exit points follow, branch to one depending on success of connection )

* connectExit -  a word terminated with a colon is used as a line label

connectExit:
* this is where we go after a successful connection.  Report to TeleFinder
* to begin logon.

CONNECT 1;
* connection established
END
* stop execution of this macro, close and return to TeleFinder

errorExit:
CONNECT 0;
* connection NOT established
CANCEL
* terminate dialing, let user take another course of action

END
* stop execution of this macro, close and return to TeleFinder

Typical Macro Body
A flow of statements that is typical to use follows.  This example shows setting up a 2400 baud 
modem, dialing a number, and testing for a connection.

*Initialize serial port to 2400 bps, 8 data bits, no parity, 1 stop bit
BAUD 2400;
SETCOMM "8N1"

* Get the modems attention with AT, delay characters by 4 ticks to
* simulate typing.  Wait 4 seconds for the OK response, continue if
* successful, re-try if not.
resetBd:
SLOW 4;
TYPE "AT^M"
PROMPSEC 4;
PROMPT "OK"
FALSE "resetBd"

* Turn on the modems speaker and send the dial command.
ReDial:
TYPE "ATM1^M"
PROMPT "OK"
TYPE "ATDT730-5785^M"

* Pause 2 seconds to clear the buffer of command echoes
PAUSE 2;

* set the wait time to 45 seconds for the connection response
PROMPSEC 45;

* put a string ending in CR into the macro buffer
* ( this is the beginning of the response ( ie: <CR><LF>CONNECT<CR><LF>)
GETLINE "^M"

* dump out if the CR never comes
FALSE "errorExit"

* put the rest of the response into the buffer
GETLINE "^M"

* dump out if the CR never comes
FALSE "errorExit"

* is CONNECT in the macro buffer?
CMPLINE "CONNECT"

* if yes, we have connected. Branch to connectExit.
TRUE "connectExit"

* CONNECT not seen, maybe it was busy?
CMPLINE "BUSY"

* if yes do reset and redial
TRUE "resetBd"
* it was not BUSY either ( we only handle BUSY and CONNECT in
* the example ) branch to errorExit.
GOTO "errorExit"

You can write more complex macros by  making more comparisons which may branch to different 
places in the script.

Notes
1)  Some connections may not report a connection status.  In this case you can prompt for 
"TeleFinder" after dialing.  The host  repeatedly sends the "TeleFinder" string after it has answered the
phone to initiate a logon. ( the serial port format should be set to "8N1" at this time to match with the 
host ).

2)  If you end a macro without setting CONNECT 1; or the CANCEL flag TeleFinder will proceed with 
its default "Modem Wake-Up" routine and dialing procedure.

3)  If the user Cancels a macro from the the "Macro Trace" dialog the cancel flag is set.

Macro Command Reference
The following is a reference for the various commands you can enter in a macro file.

legend
nnnn; indicates numeric parameter required.
"aaa" indicates character string required  by the command .

Delimit numeric parameters with a semicolon (;).  

Enclose string parameters with plain double quotes.  Do not use smart quotes.  Use the ^ to create a 
control character in a string ( ^H is the same as typing control-H, ^M is the same as typing return ).

control of serial port

BAUD nnnn;
Set the serial port data rate in bits per second.
Select from 300, 1200, 2400, 9600, 14400, 19200, 38400 or 57600.

SETCOMM "aaa"
Sets the serial port format.  The 8N1 format that TeleFinder requires is default.
"8N1" for 8 data bits, no parity, 1 stop bit.
"7E1" for 7 data bits, even parity, 1 stop bit.  

FLOW n;
Set serial port flow control.
1 No flow control
2 CTS flow control ( partial handshake )
3 CTS/DTR flow control ( hardware handshake )
4 XON/XOFF flow control

REMECHO n;
Echo received data back to sender.  Echo off is default.
1 - echo on
0 - echo off

BREAK n;
Send a serial break signal for "n" ticks  (1/60 second ).

sending data

TYPE "aaa"
Send the characters in quotes thru the serial port.  

SLOW n;
Delay n ticks between characters.  This slows the speed of the TYPE command.  There are 60 ticks 
in one second.  Some modems require this pacing because they cannot interpret commands at 
computer speeds. 

receiving and comparing data

PROMPSEC n;
Use this to shorten the amount of time the PROMPT and GETLINE commands wait for strings.  
Specify a time period in seconds.

PROMPT "aaa"
Wait for the enclosed string to be received through the serial port.  This command sets the 
TRUE/FALSE flags depending upon its success ( TRUE if aaa was seen ).  See controlling macro 
execution for more information about the TRUE and FALSE flags.

GETLINE "a"
Buffer a string ending in "a" from the characters received through the serial port.   This command sets
the TRUE/FALSE flags depending upon its success ( TRUE if a line was was buffered ).

CMPLINE "aaa"
Compare "aaa" to the string in the buffer.  TRUE if "aaa" is in the buffer, FALSE if not.  CMPLINE is 
case sensitive.

controlling macro execution

PAUSE n;
Wait n seconds and discard and data received during that time.

GOTO "label"
Execute statements following the label. Labels are any word which is not a command beginning at the
start of a line and ending with a colon.

TRUE "label"
If the TRUE flag was set by PROMPT, GETLINE, or CMPLINE goto "label".

FALSE "label"
If the FALSE  flag was set by PROMPT, GETLINE, or CMPLINE goto "label".

END
End the macro and return to TeleFinder.

setting flags and misc.

CONNECT n;
Set the connect flag to 1 or 0.  TeleFinder looks at the connect flag to determine if a connection is 
present.  TeleFinder begins the logon when the connect flag is 1.  TeleFinder begins with waking the 
modem if the connect flag is zero.

Use CONNECT 1; when you make a connection.  Use CONNECT 0; when you want TeleFinder to 
make a connection with its built in routines.

CANCEL
Set the cancel flag to 1.  Use this to cancel the dialing process.

BELL
Create the System Beep sound.

*
Precede comment lines with *.  The command parser ignores the rest of the line.

Installing the Macro Tool
Copy TF Macro Tool to the same folder as TeleFinder/User.  TeleFinder loads the Macro Tool only if it 
is in the same folder as TeleFinder when you first open TeleFinder.  It’s icon appears in the upper right
hand portion of TeleFinder’s desktop.

Open the TF Macro Tool to access this dialog.  Click in the Select a Macro button to choose a macro 
file to use.  Turn the macro tool off when you do not need to use it.  Click in the OK button to save 
your selections.  Click in the Cancel button to close the dialog without making any changes.

Only select a Macro are which is in the same folder as TeleFinder.  If the TF Macro Tool cannot  find 
the macro file the open dialog will automatically appear when the user  dials your BBS.

Select the macro tool icon and choose “Get Info” from the TeleFinder/user file menu.  This will show 
you it’s current revision number. 

Info for the Macro Tool version 1.1 B0.

